
DRAFT
This analysis used data for fiscal years 2008 and 2009, grouped using version 27.0 of the PPR grouper and used readmissions within 30 days.

PPR rates were calculated by APR-DRG/SOI using the entire data set. These statewide readmission rates were then used as the expected values in the analysis.

The actual to expected PPR rates were calculated by age category and mental health status, and the ratio of the two was used as an adjustment factor for age category and mental health status. The age categories used were 0-17, 18-64, and 65 and older. The mental health status used the flag pprmhs returned by the PPR grouper. pprmhs=3 indicates the presence of a mental health diagnosis. The actual, expect and adjustment factors were as follows:

     +--------------------------------------------------+

     | pprmhs   agecat    actrate    pprrate   adjfac~r |

     |--------------------------------------------------|

  1. |      0        0   .0304167   .0450827   .6746867 |

  2. |      3        0   .0626209   .0736083   .8507312 |

  3. |      0        1   .0479774   .0513196   .9348758 |

  4. |      3        1   .0885143   .0819156   1.080554 |

  5. |      0        3   .0939762   .0893872   1.051338 |

     |--------------------------------------------------|

  6. |      3        3   .1060019   .0960677   1.103409 |

     +--------------------------------------------------+

A chain was determined to be Medicaid if the principal or secondary payer was Medicaid or Medicaid HMO for any discharge for that patient in the data set. Using this definition of Medicaid the Medicaid patients were found to have a substantially higher PPR rate than non-Medicaid patients. The adjustment factor for Medicaid was 1.195, and for non-Medicaid was 0.94.
The program used to calculate these factors in attached as Appendix A.

Medicare, Blue Cross and Medicaid out-of-state adjustment factors
In order to adjust for out-of-state readmissions, which would be expected to be higher for hospitals close to borders with other states, Medicare data was obtained for federal fiscal year 2008. 

The rate of PPRs was calculated by hospital, along with the expected rate using the statewide expected rates developed previously using all payers, and the age and mental health adjustment factors previously listed. The ratio of the actual to the expected was calculated, first using discharges to hospitals in any state, and then using just discharges from Maryland hospitals. The ratio of these two was the adjustment factor to be applied to adjust for out-of-state Medicare readmissions. 

The program used to calculate these factors is attached as Appendix B.

This readmission factor should be combined with the corresponding  factor developed by Blue Cross to calculate an estimated adjustment factor for out-of-state readmissions.
It has not been possible to develop a similar adjustment using Medicaid data because the data from Medicaid had only CPT and not ICD procedure codes, so could not be run through the PPR grouper, and also has additional problems. 

Calculation of chain weights 
The previous and current calculations have all been based on the number of readmissions, with all readmissions weighted equally. Clearly the costs associated with readmissions will vary by the type of initial admission. The calculation described in this section expands the calculation of the relative PPR rates of the hospitals to take into account the mix of initial admissions in chains by APR-DRG and SOI. The programming to perform that analysis has not yet been completed. 
The weight for a re-admission chain was calculated by summing the APR-DRG/SOI weights for each readmission in the chain. These weights were then assigned to all readmission chains as the "actual" weight for the chain. The chain weights were then summarized by calculating the mean chain weight for all chains following an initial or only admission in a given APR-DRG/SOI. The resulting weight is the expected weight for readmissions following the initial or only admission in the particular APR-DRG/SOI. The rankings were then recalculated using these weights.
Using this methodology the age and mental health factors became:

     +-----------------------------------------+

     | pprage    actrate    pprrate   ageindex |

     |-----------------------------------------|

  1. |      B   .0506826    .093799   .5403314 |

  2. |      C   .0889933   .1122925   .7925128 |

  3. |      D   .0750117   .0944965   .7938038 |

  4. |      E   .0807381   .1069761   .7547295 |

  5. |      F   .0825727    .084213   .9805226 |

     |-----------------------------------------|

  6. |      G   .1763076   .1811897   .9730553 |

  7. |      H   .2138714   .2119366   1.009129 |

  8. |      I   .2348037   .2196361   1.069058 |

  9. |      J   .2322788   .2192625   1.059364 |

     +-----------------------------------------+

and for the consolidated age categories split by mental health status were:

   +--------------------------------------------------+

     | pprmhs   agecat    actrate    pprrate   adjfac~r |

     |--------------------------------------------------|

  1. |      0        0   .0762713   .1007334   .7571598 |

  2. |      3        0   .1267907   .1518696   .8348653 |

  3. |      0        1   .1161369   .1221828   .9505174 |

  4. |      3        1   .1790091   .1693812   1.056841 |

  5. |      0        3   .2241034    .215215     1.0413 |

     |--------------------------------------------------|

  6. |      3        3   .2516828   .2340919   1.075145 |

     +--------------------------------------------------+

As before, a chain was determined to be Medicaid if the principal or secondary payer was Medicaid or Medicaid HMO for any discharge for that patient in the data set. Using this definition of Medicaid the Medicaid patients were still found to have a substantially higher PPR rate than non-Medicaid patients. The adjustment factor for Medicaid was 1.23, and for non-Medicaid was 0.92.

The programs used to calculate these factors in attached as Appendix C.

Options for level of adjustment
The factor for which the hospitals are placed at risk has a number of different possible levels:

1) the PPR rate. This is what has been presented in previous meetings.
2) the PPR rate, weighted by the expected weight associated with chains starting with the particular APR-DRG/SOI in the initial admission. This is the method used in the preceding discussion.
3) the PPR rate, adjusted to account for the actual weight of readmissions in the subsequent chain.

4) option 3, but with some outlier threshold applied to limit the weight for which the initial hospital was accountable. 

Appendix A

*** Analysis of readmission data for 2008 and 2009 by age, 

*** mental health marker, and payer

set more off

cd c:\myfiles\ppr

use md0809_a30, clear /* Using the 2008-2009 PPR data from 3M  */

drop if prrerr~=0 /* dropping error records */

drop if (pprtype=="IA"|pprtype=="IR"|pprtype=="OA"|pprtype=="TR")&(dischstat==20|dischstat==24|dischstat==26) /* initial dropping cases transferred to onsite rehab, chronic or psych unit */

generate float provno=provider

sort uniqueid

save temp, replace

keep uniqueid pay1 pay2

*** generating a Medicaid indicator if Medicaid is present in any discharge for the uniqueid

generate mapayr=0

replace mapayr=1 if pay1==02|pay2==02|pay1==14|pay2==14

collapse (sum) mapayr, by(uniqueid)

tab mapayr

replace mapayr=1 if mapayr>1 & mapayr~=.

tab mapayr

sort uniqueid 

merge 1:m uniqueid using temp

tab _merge

drop if _merge==1

drop _merge

replace mapayr=0 if mapayr==.

replace mapayr=1 if pay1==02|pay2==02|pay1==14|pay2==14

tab mapayr

count

*** Developing denominator counts by APR DRG and SOI

keep if pprtype=="IA"|pprtype=="IR"|pprtype=="OA"|pprtype=="TR"

generate ma=0

replace ma=1 if pay1==2|pay2==02|pay1==14|pay2==14

sort pprdrg pprsoi

save temp, replace

collapse (count) prrerr, by(pprdrg pprsoi)

ren prrerr pprdenom

sort pprdrg pprsoi

save pprrates, replace

*** Developing numerator counts and readmission rates by APR DRG and SOI

use temp, clear

keep if pprtype=="IA"|pprtype=="IR"

collapse (count) prrerr, by(pprdrg pprsoi)

ren prrerr pprnum

sort pprdrg pprsoi

merge 1:m pprdrg pprsoi using pprrates

generate pprrate=pprnum/pprdenom

replace pprrate=0 if pprrate==.

drop _merge

sort pprdrg pprsoi

save pprrates, replace

*** Merging the statewide readmission rates into the data

merge 1:m pprdrg pprsoi using temp

tab _merge

drop if _merge~=3

drop _merge

generate actrate=0

replace actrate=1 if pprtype=="IA"|pprtype=="IR" /* Marking readmissions */

save temp, replace

sum actrate pprrate /* Checking the actual and expected rates are the same at the state level */

*** Analysis by age category

collapse (mean) actrate pprrate, by(pprage)

generate ageindex=actrate/pprrate

list _all

*** Analysis by mental health indicator

use temp, clear

sum actrate pprrate if pprmhs==3 

sum actrate pprrate if pprmhs~=3

*** Analysis by age category and mental health indicator

use temp, clear

collapse (mean) actrate pprrate (count) provno, by(pprage pprmhs)

ren provno casecnt

generate ageindex=actrate/pprrate

list _all

*** Adjusting for age category and mental health status

use temp, clear

*** age categories: 17 and under, 18-64, 65 and older

generate agecat=0 if pprage=="A"|pprage=="B"|pprage=="C"|pprage=="D"|pprage=="E"

replace agecat=1 if pprage=="F"|pprage=="G"

replace agecat=3 if  pprage=="J"|pprage=="H"|pprage=="I"

sort agecat pprmhs

save temp, replace

collapse (mean) actrate pprrate, by(agecat pprmhs)

generate adjfactor=actrate/pprrate

list _all

drop actrate pprrate

sort agecat pprmhs

save age_mh_factors, replace

merge 1:m agecat pprmhs using temp

drop if _merge ~=3

drop _merge

generate adjexpect=pprrate*adjfactor

save temp, replace

***Analysis by payer

use temp, clear

sort mapayr

by mapayr: sum actrate adjexpect

sort ma

by ma: sum actrate adjexpect

tab ma mapayr

*** Analysis by hospital with adjustment for age, DSH, Medicaid in chain, and mental health

replace adjexpect=adjexpect*0.94 if mapayr==0

replace adjexpect=adjexpect*1.195 if mapayr==1

collapse (mean) actrate adjexpect, by(provno)

generate index=actrate/adjexpect

sort provno

merge provno using names_ids

drop if _merge ~=3

drop _merge

sort index

list name provno index actrate adjexpect

save ppr_results_2008_9, replace

Appendix B
*** Analysis of MedPAR readmission data for 2008 with adjustment for age and mental health marker

set more off

cd c:\myfiles\ppr

use medpar_all_a30, clear /* Using the all state Medpar PPR grouped data */

drop if prrerr~=0 /* dropping error records */

drop if (pprtype=="IA"|pprtype=="IR"|pprtype=="OA"|pprtype=="TR")&(dischstat==20|dischstat==24|dischstat==26) /* initial dropping cases transferred to onsite rehab, chronic or psych unit */

generate float provno=provider

sort pprdrg pprsoi

*** Merging the statewide readmission rates into the data

merge m:1 pprdrg pprsoi using pprrates

tab _merge

drop if _merge~=3

drop _merge

generate actrate=0

replace actrate=1 if pprtype=="IA"|pprtype=="IR" /* Marking readmissions */

sum actrate pprrate /* comparing the actual and expected rates are the same at the state level */

*** Adjusting for age category and mental health status

*** age categories: 17 and under, 18-64, 65 and older

generate agecat=0 if pprage=="A"|pprage=="B"|pprage=="C"|pprage=="D"|pprage=="E"

replace agecat=1 if pprage=="F"|pprage=="G"

replace agecat=3 if  pprage=="J"|pprage=="H"|pprage=="I"

sort agecat pprmhs

save temp, replace

use age_mh_factors, clear /* using the Maryland statewide age and mental health adjustment factors */

sort agecat pprmhs

merge 1:m agecat pprmhs using temp

drop if _merge ~=3

drop _merge

generate adjexpect=pprrate*adjfactor

save temp, replace

*** Summary by hospital after adjustment for age and mental health

collapse (mean) actrate adjexpect, by(provno)

generate index_all=actrate/adjexpect

ren actrate actrate_all

ren adjexpect expt_all

sort provno

merge provno using names_ids

drop if _merge ~=3

drop _merge

sort provno

save medicare_adjs, replace

*** Doing the analysis with Maryland discharges only 

use medpar_md_a30, clear /* Using the all state Medpar PPR grouped data */

drop if prrerr~=0 /* dropping error records */

drop if (pprtype=="IA"|pprtype=="IR"|pprtype=="OA"|pprtype=="TR")&(dischstat==20|dischstat==24|dischstat==26) /* initial dropping cases transferred to onsite rehab, chronic or psych unit */

generate float provno=provider

sort pprdrg pprsoi

*** Merging the statewide readmission rates into the data

merge m:1 pprdrg pprsoi using pprrates

tab _merge

drop if _merge~=3

drop _merge

generate actrate=0

replace actrate=1 if pprtype=="IA"|pprtype=="IR" /* Marking readmissions */

sum actrate pprrate /* comparing the actual and expected rates are the same at the state level */

*** Adjusting for age category and mental health status

*** age categories: 17 and under, 18-64, 65 and older

generate agecat=0 if pprage=="A"|pprage=="B"|pprage=="C"|pprage=="D"|pprage=="E"

replace agecat=1 if pprage=="F"|pprage=="G"

replace agecat=3 if  pprage=="J"|pprage=="H"|pprage=="I"

sort agecat pprmhs

save temp, replace

use age_mh_factors, clear /* using the Maryland statewide age and mental health adjustment factors */

sort agecat pprmhs

merge 1:m agecat pprmhs using temp

drop if _merge ~=3

drop _merge

generate adjexpect=pprrate*adjfactor

save temp, replace

*** Summary by hospital after adjustment for age and mental health

collapse (mean) actrate adjexpect, by(provno)

generate index_md=actrate/adjexpect

ren actrate actrate_md

ren adjexpect expt_md

sort provno

merge provno using medicare_adjs

drop if _merge ~=3

drop _merge

generate mc_adj_factor=index_all/index_md

sort mc_adj_factor

Appendix C
*** ppr_chain_wgts.do, calculating PPR chain weights for 2008 and 2009 

set more off

cd c:\myfiles\ppr

use md0809_a30, clear /* Using the 2008-2009 PPR data from 3M  */

drop if prrerr~=0 /* dropping error records */

drop if (pprtype=="IA"|pprtype=="IR"|pprtype=="OA"|pprtype=="TR")&(dischstat==20|dischstat==24|dischstat==26) /* initial dropping cases transferred to onsite rehab, chronic or psych unit */

generate float provno=provider

keep if pprtype=="TR"|pprtype=="IA"|pprtype=="IR"|pprtype=="RA"|pprtype=="RT"|pprtype=="OA"

sort uchainno

save temp, replace

drop if pprtype=="OA"|pprtype=="TR"

sort pprdrg pprsoi

merge m:1 pprdrg pprsoi using apr_weights /* merging in APR-DRG weights for FY 2010 */

drop if _merge ~= 3

replace weight=0 if pprtype=="IA"|pprtype=="IR"

collapse (sum) weight, by(uchainno)

drop if uchainno==0

sort uchainno

save chain_weights, replace

merge 1:m uchainno using temp

drop if _merge==1

drop _merge

sum weight, detail

replace weight=0 if weight==.

sum weight

keep if pprtype=="IA"|pprtype=="OA"|pprtype=="TR"|pprtype=="IR"

sum weight

save temp, replace

collapse (mean) weight, by(pprdrg pprsoi)

sort pprdrg pprsoi

save ppr_ppr_weights, replace

use temp

drop if weight==0

drop if pprtype=="OA"|pprtype=="TR"

sum weight, detail

collapse (mean) weight, by(pprdrg pprsoi)

sort pprdrg pprsoi

save ppr_chain_weights, replace
*** Analysis of readmission data for 2008 and 2009 by age, 

*** mental health marker, and payer

set more off

cd c:\myfiles\ppr

use md0809_a30, clear /* Using the 2008-2009 PPR data from 3M  */

drop if prrerr~=0 /* dropping error records */

drop if (pprtype=="IA"|pprtype=="IR"|pprtype=="OA"|pprtype=="TR")&(dischstat==20|dischstat==24|dischstat==26) /* initial dropping cases transferred to onsite rehab, chronic or psych unit */

generate float provno=provider

keep if pprtype=="TR"|pprtype=="IA"|pprtype=="IR"|pprtype=="OA"

sort uniqueid

save temp, replace

keep uniqueid pay1 pay2

*** generating a Medicaid indicator if Medicaid is present in any discharge for the uniqueid

generate mapayr=0

replace mapayr=1 if pay1==02|pay2==02|pay1==14|pay2==14

collapse (sum) mapayr, by(uniqueid)

tab mapayr

replace mapayr=1 if mapayr>1 & mapayr~=.

tab mapayr

sort uniqueid 

merge 1:m uniqueid using temp

tab _merge

drop if _merge==1

drop _merge

replace mapayr=0 if mapayr==.

replace mapayr=1 if pay1==02|pay2==02|pay1==14|pay2==14

tab mapayr

count

*** Developing expected weights by APR DRG and SOI

keep if pprtype=="IA"|pprtype=="IR"|pprtype=="OA"|pprtype=="TR"

generate ma=0

replace ma=1 if pay1==2|pay2==02|pay1==14|pay2==14

sort pprdrg pprsoi

merge m:1 pprdrg pprsoi using ppr_ppr_weights

drop if _merge==2

drop _merge

replace weight=0 if weight==.

ren weight pprrate

sort pprdrg pprsoi

save temp, replace

*** Developing numerator readmission weights by APR DRG and SOI

sort pprdrg pprsoi

merge m:1 pprdrg pprsoi using ppr_chain_weights

drop if _merge==2

drop _merge

replace weight = 0 if weight==.

replace weight=0 if pprtype=="OA"|pprtype=="TR"

ren weight actrate

sort pprdrg pprsoi

save pprrates, replace

save temp, replace

sum actrate pprrate /* Checking the actual and expected rates are the same at the state level */

*** Analysis by age category

collapse (mean) actrate pprrate, by(pprage)

generate ageindex=actrate/pprrate

list _all

*** Analysis by mental health indicator

use temp, clear

sum actrate pprrate if pprmhs==3 

sum actrate pprrate if pprmhs~=3

*** Analysis by age category and mental health indicator

use temp, clear

collapse (mean) actrate pprrate (count) provno, by(pprage pprmhs)

ren provno casecnt

generate ageindex=actrate/pprrate

list _all

*** Adjusting for age category and mental health status

use temp, clear

*** age categories: 17 and under, 18-64, 65 and older

generate agecat=0 if pprage=="A"|pprage=="B"|pprage=="C"|pprage=="D"|pprage=="E"

replace agecat=1 if pprage=="F"|pprage=="G"

replace agecat=3 if  pprage=="J"|pprage=="H"|pprage=="I"

sort agecat pprmhs

save temp, replace

collapse (mean) actrate pprrate, by(agecat pprmhs)

generate adjfactor=actrate/pprrate

list _all

drop actrate pprrate

sort agecat pprmhs

save age_mh_factors, replace

merge 1:m agecat pprmhs using temp

drop if _merge ~=3

drop _merge

generate adjexpect=pprrate*adjfactor

save temp, replace

***Analysis by payer

use temp, clear

sort mapayr

by mapayr: sum actrate adjexpect

sort ma

by ma: sum actrate adjexpect

tab ma mapayr

*** Analysis by hospital with adjustment for age, DSH, Medicaid in chain, and mental health

replace adjexpect=adjexpect*0.92 if mapayr==0

replace adjexpect=adjexpect*1.23 if mapayr==1

collapse (sum) actrate adjexpect, by(provno)

generate index=actrate/adjexpect

sort provno

merge provno using names_ids

drop if _merge ~=3

drop _merge

sort provno

save ppr_results_2008_9, replace

*** generating case mix weight for included cases by hospital

use temp, clear

sort pprdrg pprsoi

merge m:1 pprdrg pprsoi using apr_weights

drop if _merge ~=3

drop _merge

collapse (sum) weight, by(provno)

ren weight totcmw

sort provno

merge 1:1 provno using ppr_results_2008_9

drop if _merge ~=3

drop _merge

generate alloc_basis=(actrate-adjexpect)/totcmw

sort alloc_basis

save ppr_results_2008_9, replace

